
Corrections required for test case 8.5.4.1

Change 1

	Testcase name
	8.5.4.1

	Reason for change
	According to prose 36.523-1-a00 section 8.5.4.1.3.2 in Step 19a1 SS transmits a UECapabilityEnquiry message to request UE radio access capability information for every other supported RAT. 

However in the step 19a1 in TTCN the message UEcapabilityEnquiry is sent with all capability requested (utran, geran-ps, geran-cs or cdma2000-1xRTT). According to 36.523-1-a00 Table 8.5.4.1.3.3-3 UEcapabilityEnquiry should contain capabilities only according to value of PICS parameters.

	Summary of change
	Created new template cs_SuppOtherRatCapReq, with one integer variable and added three IF statements where the cs_SuppOtherRatCapReq is filled in. In the sending section template cs_AllOtherRatCapReq is replaced with the new one.

	Source of change
	\8_5\ RRC_Others.ttcn

	MCC160 Comment
	Accepted


Before change:

	function fl_AllRatCapabilityCheck ( CellId_Type p_CellId,

                                      RRC_TransactionIdentifier p_RRC_TI ) runs on EUTRA_PTC

  {

    var SRB_COMMON_IND v_ReceivedAspForUeCapabilityInfo;

    //* @desc Fill the UECapabilityInformation_r8_IEs with expected RAT-containers based on set PICS parameters

    var template UECapabilityInformation_r8_IEs v_OtherSupRatCapability := fl_SetUeOtherSupRatCap ( );

    //* @desc Transmit UE Capability Enquiry message for every other RAT

    SRB.send ( cas_SRB1_RrcPdu_REQ ( p_CellId,

                                     cs_TimingInfo_Now,

                                     cs_508_UeCapabilityEnquiry ( p_RRC_TI,

                                                                  cs_AllOtherRatCapReq ) ) );

    


After change:

	    function fl_AllRatCapabilityCheck ( CellId_Type p_CellId,

                                      RRC_TransactionIdentifier p_RRC_TI ) runs on EUTRA_PTC

  {

    var SRB_COMMON_IND v_ReceivedAspForUeCapabilityInfo;

    //* @desc Fill the UECapabilityInformation_r8_IEs with expected RAT-containers based on set PICS parameters

    var template UECapabilityInformation_r8_IEs v_OtherSupRatCapability := fl_SetUeOtherSupRatCap ( );

    var template UE_CapabilityRequest cs_SuppOtherRatCapReq;                                     

    var integer i:=0;
    if ( pc_GERAN and pc_CS){                                     

    cs_SuppOtherRatCapReq[i]:= geran_cs;

      i:=i+1;   

    }

    if ( pc_GERAN and pc_PS  ) {

      cs_SuppOtherRatCapReq[i]:= geran_ps;

      i:=i+1; 

    }    

   if ( pc_FDD or pc_TDD_LCR or pc_TDD_HCR or pc_TDD_VHCR ) {    

   cs_SuppOtherRatCapReq[i]:=utra;

      i:=i+1;

     }

    if ( pc_HRPD or pc_1xRTT ){                                                    

      cs_SuppOtherRatCapReq[i]:= cdma2000_1XRTT;                           

      i:=i+1; 

    }
    //* @desc Transmit UE Capability Enquiry message for every other RAT

    SRB.send ( cas_SRB1_RrcPdu_REQ ( p_CellId,

                                     cs_TimingInfo_Now,

                                     cs_508_UeCapabilityEnquiry ( p_RRC_TI,

                                                                  cs_SuppOtherRatCapReq ) ) );                  


Change 2

	Testcase name
	8.5.4.1

	Reason for change
	According to 36.523-1-a00 section 8.5.4.1.3.2 in Step 19a2 UE should transmits a UECapabilityInformation message containing capabilities depending on what has been requested (Table 8.5.4.1.3.3-4).
However TTCN does not distinguish between PS and CS capabilities in case of GERAN.

	Summary of change
	Split one IF statement to two IF statements to distinguish between PS and CS GERAN cases.

	Source of change
	\8_5\ RRC_Others.ttcn

	MCC160 Comment
	Accepted


Before change:

	function fl_SetUeOtherSupRatCap ( ) return template UECapabilityInformation_r8_IEs

  {

    var template (present) UE_CapabilityRAT_ContainerList v_UE_CapabilityRAT_ContainerList;

    var integer i := 0;

    if ( pc_GERAN ){

      v_UE_CapabilityRAT_ContainerList[i] := cr_UE_CapabilityRAT_Container(geran_cs);

      i := i + 1;

      v_UE_CapabilityRAT_ContainerList[i] := cr_UE_CapabilityRAT_Container(geran_ps);

      i := i + 1;

    }
    if ( pc_FDD or pc_TDD_LCR or pc_TDD_HCR or pc_TDD_VHCR ) {

      v_UE_CapabilityRAT_ContainerList[i] := cr_UE_CapabilityRAT_Container(utra);

      i := i + 1;

    }

    if ( pc_HRPD or pc_1xRTT ){

      v_UE_CapabilityRAT_ContainerList[i] := cr_UE_CapabilityRAT_Container(cdma2000_1XRTT);

      i := i + 1;

    }

    


After change:

	function fl_SetUeOtherSupRatCap ( ) return template UECapabilityInformation_r8_IEs

  {

    var template (present) UE_CapabilityRAT_ContainerList v_UE_CapabilityRAT_ContainerList;

    var integer i := 0;

    if ( pc_GERAN and pc_CS ){                            

      v_UE_CapabilityRAT_ContainerList[i] := cr_UE_CapabilityRAT_Container(geran_cs);

      i := i + 1;

    }

    if ( pc_GERAN and pc_PS ){                            

      v_UE_CapabilityRAT_ContainerList[i] := cr_UE_CapabilityRAT_Container(geran_ps);

      i := i + 1;

    }
    if ( pc_FDD or pc_TDD_LCR or pc_TDD_HCR or pc_TDD_VHCR ) {

      v_UE_CapabilityRAT_ContainerList[i] := cr_UE_CapabilityRAT_Container(utra);

      i := i + 1;

    }

    if ( pc_HRPD or pc_1xRTT ){

      v_UE_CapabilityRAT_ContainerList[i] := cr_UE_CapabilityRAT_Container(cdma2000_1XRTT);

      i := i + 1;

    }


Additional changes from MCC160:
Due to the baseline change the test case was modified to cover Rel 9 functionality as well.
Before change:

…
  template (value) UE_CapabilityRequest cs_AllOtherRatCapReq := {utra, geran_cs, geran_ps, cdma2000_1XRTT};    /* @status    APPROVED */
…
  /*
   * @desc      Function to match the received UECapabilityInformation with the expected UECapabilityInformation.
   *            Reference 36.523-1 Table 8.5.4.1.3.3-2
   * @param     p_Octetstring                    ..  received ueCapabilitiesRAT-Container
   * @param     p_Exptected_Ue_Eutra_Capability  ..  expected ueCapabilitiesRAT-Container (set by function fl_EUTRA_SetUeEutraCapability )
   * @status    APPROVED
   */
  function fl_EUTRA_MatchEutraCapMsgs ( UE_EUTRA_Capability p_ReceivedCapabilityMsg,
                                        template UE_EUTRA_Capability p_Exptected_Ue_Eutra_Capability ) runs on EUTRA_PTC
  {
    var UE_EUTRA_Capability v_ReceivedCapabilityMsg;
    var integer i, k, l;
    var B43_Type v_NrOfSuppEutraBandsBitstring := fl_GetNrOfSuppEutraBandsBitstring();
    var charstring v_FailedEutraBand;
    var boolean v_BandSuppMatch := false;
    v_ReceivedCapabilityMsg := p_ReceivedCapabilityMsg;
    //* @desc Check that number of supported EUTRA bands match the set PICS parameters (pc_eBanda_Supp)
    if ( lengthof ( v_ReceivedCapabilityMsg.rf_Parameters.supportedBandListEUTRA ) != fl_GetNrOfSuppEutraBands() ) {
      //* @verdict fail Number of Supported EUTRA Bands according to PICS parameters
      // does not match number of entries in SupportedBandListEUTRA
      f_EUTRA_SetVerdictFailOrInconc (__FILE__, __LINE__, "Test Case 8.5.4.1 Step 10/18 - Missmatch between pc_eBanda_Supp and number of reported supported EUTRA bands");
    }
    //* @desc Check that any value B such that pc_eBandB_Supp is TRUE and different from all eutra-Band[k] where k = 1 to a - 1
    //* Go through the received supportedBandListEUTRA and match against the supported bands according to the PICS (received from fl_GetNrOfSuppEutraBandsBitstring).
    for ( k := 0; k < lengthof ( v_ReceivedCapabilityMsg.rf_Parameters.supportedBandListEUTRA ); k := k + 1 ){
      v_BandSuppMatch := false;
      for ( l := 0; l < lengthof ( v_NrOfSuppEutraBandsBitstring ); l := l + 1 ){
        //* @desc For each supported band (according to the set PICS) match the band with what is reported according to k.
        if ( (l+1) == v_ReceivedCapabilityMsg.rf_Parameters.supportedBandListEUTRA[k].bandEUTRA ){
          if ( v_NrOfSuppEutraBandsBitstring[l] == '0'B ){
            v_FailedEutraBand := int2char ( l + 1 );
            //* @verdict fail PICS setting does not match reported capability for band nr. l+1
            // or band-support har been reported more than once
            f_EUTRA_SetVerdictFailOrInconc (__FILE__, __LINE__, v_FailedEutraBand);
          }
          else {
            //* UE reports support of band v_ReceivedCapabilityMsg.rf_Parameters.supportedBandListEUTRA[k].bandEUTRA
            // and the correspponing PICS parameter is also set.
            v_NrOfSuppEutraBandsBitstring[l] := '0'B;
            v_BandSuppMatch := true;
          }
        }
      }
      if ( v_BandSuppMatch == false ) {
        //* @verdict fail UE reports support of band v_ReceivedCapabilityMsg.rf_Parameters.supportedBandListEUTRA[k].bandEUTRA
        // but corresponding PICS is not set to true.
        v_FailedEutraBand := int2char ( v_ReceivedCapabilityMsg.rf_Parameters.supportedBandListEUTRA[k].bandEUTRA );
        f_EUTRA_SetVerdictFailOrInconc (__FILE__, __LINE__, v_FailedEutraBand);
      }
    }
    //* @desc interFreqEUTRA-BandList SEQUENCE (SIZE (1..maxBands)) OF SEQUENCE {
    //* same number of entries like in supportedEUTRA-BandList
    //* Match number of entries in interFreqEUTRA-BandList with number of entries in supportedEUTRA-BandList
    for ( i := 0; i < lengthof ( v_ReceivedCapabilityMsg.measParameters.bandListEUTRA ); i := i + 1 ) {
      if ( lengthof ( v_ReceivedCapabilityMsg.rf_Parameters.supportedBandListEUTRA ) != lengthof ( v_ReceivedCapabilityMsg.measParameters.bandListEUTRA[i].interFreqBandList ) )
      {
        //* @verdict fail Number of interFreqEUTRA-BandList does not match nr. of supportedEUTRA-BandList
        f_EUTRA_SetVerdictFailOrInconc (__FILE__, __LINE__, "Test Case 8.5.4.1 Step 10/18 - Number of interFreqEUTRA-BandList does not match nr. of supportedEUTRA-BandList");
      }
    }
    //@sic R5s100302 Change 6 : Correction of check of supported interRat bands sic@
    // Check that number of bands in interRAT-Parameters matches number of entries in interRAT-BandList
    // According to 36.331:
    // interRAT-BandList - One entry corresponding to each supported band of another RAT
    //                     listed in the same order as in the interRAT-Parameters.
    for ( i := 0; i < lengthof ( v_ReceivedCapabilityMsg.measParameters.bandListEUTRA ); i := i + 1 ) {
      if ( ispresent ( v_ReceivedCapabilityMsg.measParameters.bandListEUTRA[i].interRAT_BandList ) )
      {
        if ( lengthof ( v_ReceivedCapabilityMsg.measParameters.bandListEUTRA[i].interRAT_BandList ) != fl_Count_Supported_interRAT_Bands(v_ReceivedCapabilityMsg))
        {
          //* @verdict fail Number of elements in interRAT-Parameters does not match number of entires in interRAT-BandList
          f_EUTRA_SetVerdictFailOrInconc (__FILE__, __LINE__, "Test Case 8.5.4.1 Step 10/18 - Number of listed bands in interRAT-Parameters does not match number of entires in interRAT-BandList");
        }
      }
    }
    //* @desc Match the received decoded template with the expected template
    if ( not match ( v_ReceivedCapabilityMsg, p_Exptected_Ue_Eutra_Capability )){
      //* @verdict fail Received decoded template does not match with the expected template
      f_EUTRA_SetVerdictFailOrInconc (__FILE__, __LINE__, "Test Case 8.5.4.1 Step 10/18 - Received decoded template does not match with the expected template");
    }
  }
…
  function fl_EUTRA_FullCapabilityCheck ( CellId_Type p_CellId,
                                          RRC_TransactionIdentifier p_RRC_TI ) runs on EUTRA_PTC
  {
    var SRB_COMMON_IND v_ReceivedAspForUeCapabilityInfo;
    var UE_EUTRA_Capability v_ReceivedCapabilityMsg;
    var boolean v_Result_r8, v_Result_r9;
    //* @desc Fill the IE interRAT-Parameters with expected contents based on set PICS parameters
    var template UE_EUTRA_Capability v_EutraCapability := fl_EUTRA_SetExpectedUeEutraCap ( cr_UeCapabilitiesRAT_Container );
    //* @desc Transmit UE Capability Enquiry message to ask for EUTRA Capability information
    SRB.send ( cas_SRB1_RrcPdu_REQ ( p_CellId, cs_TimingInfo_Now, cs_508_UeCapabilityEnquiry ( p_RRC_TI ) ) );
    //* @desc Receive UE Capability Information message for EUTRA capabilities and decode
    SRB.receive ( car_SRB1_RrcPdu_IND ( p_CellId, cr_508_UeCapabilityInformation ( p_RRC_TI, cr_EutraCapInfo ) ) )
    -> value v_ReceivedAspForUeCapabilityInfo;
    v_ReceivedCapabilityMsg := f_EUTRA_DecodeEutraCapMsg ( v_ReceivedAspForUeCapabilityInfo.Signalling.Rrc.Dcch.message_.c1.ueCapabilityInformation.criticalExtensions.c1.ueCapabilityInformation_r8.ue_CapabilityRAT_ContainerList[0].ueCapabilityRAT_Container );
    //* @desc Decode and check the ueCapabilitiesRAT-Container
    fl_EUTRA_MatchEutraCapMsgs ( v_ReceivedCapabilityMsg, v_EutraCapability );
    //* @desc Check consistency of FGI usage vs. release version as per 36.523-2 // @sic R5-120748 sic@
    //* @desc Rel 8 consistency checked unconditionally. Result not exploited further yet.
    v_Result_r8 := fl_CheckFeatrGrpCapFromPics_r8 ();
    //* @desc Rel 9 consistency checked if accessStratumRelease indicates Rel 9. Result not exploited further yet.
    if ( v_ReceivedCapabilityMsg.accessStratumRelease == rel9 )
    {
      v_Result_r9 := fl_CheckFeatrGrpCapFromPics_r9 ();
    };
  } // fl_EUTRA_FullCapabilityCheck
  /*
   * @desc      Send UECapabilityEnquiry to request UE radio access capability information for other supported RATs
   *            and check the received UECapabilityInformation.
   * @param     p_CellId
   * @param     p_RRC_TI
   * @status    APPROVED
   */
  function fl_AllRatCapabilityCheck ( CellId_Type p_CellId,
                                      RRC_TransactionIdentifier p_RRC_TI ) runs on EUTRA_PTC
  {
    var SRB_COMMON_IND v_ReceivedAspForUeCapabilityInfo;
    //* @desc Fill the UECapabilityInformation_r8_IEs with expected RAT-containers based on set PICS parameters
    var template UECapabilityInformation_r8_IEs v_OtherSupRatCapability := fl_SetUeOtherSupRatCap ( );
    //* @desc Transmit UE Capability Enquiry message for every other RAT
    SRB.send ( cas_SRB1_RrcPdu_REQ ( p_CellId,
                                     cs_TimingInfo_Now,
                                     cs_508_UeCapabilityEnquiry ( p_RRC_TI,
                                                                  cs_AllOtherRatCapReq ) ) );
    //* @desc Receive UE Capability Information message for every other supported RAT
    SRB.receive ( car_SRB1_RrcPdu_IND ( p_CellId, cr_508_UeCapabilityInformation ( p_RRC_TI, v_OtherSupRatCapability ) ) )
      -> value v_ReceivedAspForUeCapabilityInfo;
    //* @desc Decode and check the ueCapabilitiesRAT-Containers for every other supported RAT
    fl_EUTRA_CheckOtherRatCap ( v_ReceivedAspForUeCapabilityInfo.Signalling.Rrc.Dcch.message_.c1.ueCapabilityInformation );
  }
…
After change:

…
  template UE_EUTRA_Capability_v920_IEs cr_Ue_EUTRA_Capability_v920_IEs :=
  {
    phyLayerParameters_v920 := {
      enhancedDualLayerFDD_r9   := *,
      enhancedDualLayerTDD_r9   := *
    },
    interRAT_ParametersGERAN_v920 := {
      dtm_r9 := *,
      e_RedirectionGERAN_r9 := *
    },
    interRAT_ParametersUTRA_v920 := *,
    interRAT_ParametersCDMA2000_v920 := *,
    deviceType_r9 := *,
    csg_ProximityIndicationParameters_r9 := {
      intraFreqProximityIndication_r9 := *,
      interFreqProximityIndication_r9 := *,
      utran_ProximityIndication_r9 := *
    },
    neighCellSI_AcquisitionParameters_r9 := {
      intraFreqSI_AcquisitionForHO_r9 := *,
      interFreqSI_AcquisitionForHO_r9 := *,
      utran_SI_AcquisitionForHO_r9 := *
    },
    son_Parameters_r9  := {
      rach_Report_r9 := *
    },
    nonCriticalExtension := *
  };
…    
…

  /*
   * @desc      convert the PICS parameters pc_FeatrGrp_X_F from single booleans to a bitstring of size 32.
   *            Size 32 is chosen because the signaled element FDD featureGroupIndicators in UE-EUTRA-Capability is of size 32.
   * @return    B32_Type
   * @status    APPROVED
   */
  function fl_DeriveFeatrGrpCapFromPics_r9_FDD () return B32_Type
  {
    var B32_Type v_FeatureGroupIndicators := f_ConvertBoolToBit(pc_FeatrGrp_1_F) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_2_F) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_3_F) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_4_F) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_5_F) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_6_F) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_7_F) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_8_F) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_9_F) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_10_F) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_11_F) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_12_F) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_13_F) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_14_F) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_15_F) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_16_F) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_17_F) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_18_F) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_19_F) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_20_F) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_21_F) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_22_F) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_23_F) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_24_F) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_25_F) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_26_F) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_27_F) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_28_F) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_29_F) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_30_F) &
                                             '00'B;
    return v_FeatureGroupIndicators;
  }
  /*
   * @desc      convert the PICS parameters pc_FeatrGrp_X_T from single booleans to a bitstring of size 32.
   *            Size 32 is chosen because the signaled element TDD featureGroupIndicators in UE-EUTRA-Capability is of size 32.
   * @return    B32_Type
   * @status    APPROVED
   */
  function fl_DeriveFeatrGrpCapFromPics_r9_TDD () return B32_Type
  {
    var B32_Type v_FeatureGroupIndicators := f_ConvertBoolToBit(pc_FeatrGrp_1_T) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_2_T) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_3_T) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_4_T) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_5_T) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_6_T) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_7_T) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_8_T) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_9_T) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_10_T) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_11_T) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_12_T) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_13_T) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_14_T) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_15_T) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_16_T) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_17_T) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_18_T) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_19_T) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_20_T) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_21_T) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_22_T) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_23_T) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_24_T) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_25_T) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_26_T) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_27_T) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_28_T) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_29_T) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_30_T) &
                                             '00'B;
    return v_FeatureGroupIndicators;
  }
  /*
   * @desc      convert the PICS parameters additional pc_FeatrGrp_X additional from single booleans to a bitstring of size 32.
   *            Size 32 is chosen because the signaled element additional featureGroupIndicators in UE-EUTRA-Capability is of size 32.
   * @return    B32_Type
   * @status    APPROVED
   */
  function fl_DeriveFeatrGrpCapFromPics_r9Add () return B32_Type
  {
    var B32_Type v_FeatureGroupIndicators := f_ConvertBoolToBit(pc_FeatrGrp_33) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_34) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_35) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_36) &
                                             '0000000000000000000000000000'B;
    return v_FeatureGroupIndicators;
  }
  /*
   * @desc      convert the PICS parameters additional pc_FeatrGrp_X_F from single booleans to a bitstring of size 32.
   *            Size 32 is chosen because the signaled element additional featureGroupIndicators FDD in UE-EUTRA-Capability is of size 32.
   * @return    B32_Type
   * @status    APPROVED
   */
  function fl_DeriveFeatrGrpCapFromPics_r9Add_FDD () return B32_Type
  {
    var B32_Type v_FeatureGroupIndicators := f_ConvertBoolToBit(pc_FeatrGrp_33_F) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_34_F) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_35_F) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_36_F) &
                                             '0000000000000000000000000000'B;
    return v_FeatureGroupIndicators;
  }
  /*
   * @desc      convert the PICS parameters additional pc_FeatrGrp_X_T from single booleans to a bitstring of size 32.
   *            Size 32 is chosen because the signaled element additional featureGroupIndicators TDD in UE-EUTRA-Capability is of size 32.
   * @return    B32_Type
   * @status    APPROVED
   */
  function fl_DeriveFeatrGrpCapFromPics_r9Add_TDD () return B32_Type
  {
    var B32_Type v_FeatureGroupIndicators := f_ConvertBoolToBit(pc_FeatrGrp_33_T) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_34_T) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_35_T) &
                                             f_ConvertBoolToBit(pc_FeatrGrp_36_T) &
                                             '0000000000000000000000000000'B;
    return v_FeatureGroupIndicators;
  }
  //=========================================================================
  /*
   * @desc      check the restrictions on FGI and other PICS as per 36.523-2 related to Release 8
   *            inconsistencies lead to verdict INCONCLUSIVE
   * @status    APPROVED
   */
  function fl_CheckFeatrGrpCapFromPics_r8 ()
  { // @sic R5-120748 sic@
    // 1. mandatory PICS - none
    // 2. relationship between PICS
    // 2.1 relationships between FGIs
    if (pc_FeatrGrp_3 and not pc_FeatrGrp_7)                  {f_EUTRA_SetVerdictInconc(__FILE__, __LINE__, "FGI 3 set incorrectly");}
    if (pc_FeatrGrp_4 and not pc_FeatrGrp_5)                  {f_EUTRA_SetVerdictInconc(__FILE__, __LINE__, "FGI 4 set incorrectly");}
    // - can only be set to 0 if the UE does not support voice; for the time being VoLTE is used
    if (not pc_FeatrGrp_7 and pc_VoLTE)                       {f_EUTRA_SetVerdictInconc(__FILE__, __LINE__, "FGI 7 set incorrectly");}
    if (pc_FeatrGrp_8 and not pc_FeatrGrp_22)                 {f_EUTRA_SetVerdictInconc(__FILE__, __LINE__, "FGI 8 set incorrectly");}
    if (pc_FeatrGrp_9 and not pc_FeatrGrp_23)                 {f_EUTRA_SetVerdictInconc(__FILE__, __LINE__, "FGI 9 set incorrectly");}
    if (pc_FeatrGrp_11 and not pc_FeatrGrp_24)                {f_EUTRA_SetVerdictInconc(__FILE__, __LINE__, "FGI 11 set incorrectly");}
    if (pc_FeatrGrp_12 and not pc_FeatrGrp_26)                {f_EUTRA_SetVerdictInconc(__FILE__, __LINE__, "FGI 12 set incorrectly");}
    if (pc_FeatrGrp_13 and not pc_FeatrGrp_25)                {f_EUTRA_SetVerdictInconc(__FILE__, __LINE__, "FGI 13 set incorrectly");}
    if (pc_FeatrGrp_15 and not pc_FeatrGrp_22 and not pc_FeatrGrp_23 and not pc_FeatrGrp_24 and not pc_FeatrGrp_26)  {f_EUTRA_SetVerdictInconc(__FILE__, __LINE__, "FGI 15 set incorrectly");}
    if (pc_FeatrGrp_17 and not pc_FeatrGrp_5)                 {f_EUTRA_SetVerdictInconc(__FILE__, __LINE__, "FGI 17 set incorrectly");}
    if (pc_FeatrGrp_18 and not pc_FeatrGrp_5)                 {f_EUTRA_SetVerdictInconc(__FILE__, __LINE__, "FGI 18 set incorrectly");}
    if (pc_FeatrGrp_19 and not pc_FeatrGrp_22 and not pc_FeatrGrp_23 and not pc_FeatrGrp_24 and not pc_FeatrGrp_26)  {f_EUTRA_SetVerdictInconc(__FILE__, __LINE__, "FGI 19 set incorrectly");}
    // - Regardless of what bit number 7 and bit number 20 is set to, UE shall support at least SRB1 and SRB2 for DCCH + 4x AM DRB
    // - Regardless of what bit number 20 is set to, if bit number 7 is set to '1', UE shall support at least SRB1 and SRB2 for DCCH + 4x AM DRB + 1x UM DRB
    // if  (pc_FeatrGrp_20 == ? and '???')  {f_EUTRA_SetVerdictInconc(__FILE__, __LINE__, "FGI 20 set incorrectly");}
    if (pc_FeatrGrp_27 and not pc_FeatrGrp_8)                 {f_EUTRA_SetVerdictInconc(__FILE__, __LINE__, "FGI 27 set incorrectly");}
    if (pc_FeatrGrp_30 and not pc_FeatrGrp_13)                {f_EUTRA_SetVerdictInconc(__FILE__, __LINE__, "FGI 30 set incorrectly");}
    // 2.2 other relationships
    if (pc_FeatrGrp_3 and not pc_Semi_Persistence_Scheduling) {f_EUTRA_SetVerdictInconc(__FILE__, __LINE__, "pc_Semi_Persistence_Scheduling set incorrectly(R8)");}
    if (pc_FeatrGrp_3 and not pc_TTI_Bundling)                {f_EUTRA_SetVerdictInconc(__FILE__, __LINE__, "pc_TTI_Bundling set incorrectly(R8)");}
  } // fl_CheckFeatrGrpCapFromPics_r8
  /*
   * @desc      check the restrictions on FGI and other PICS as per 36.523-2 related to Release 9
   *            inconsistencies lead to verdict INCONCLUSIVE
   * @status    APPROVED
   */
  function fl_CheckFeatrGrpCapFromPics_r9 ()
  { // @sic R5-120748 sic@
    var boolean v_Band13OnlySupported := pc_eBand13_Supp and
      not pc_eBand1_Supp  and not pc_eBand2_Supp  and not pc_eBand3_Supp  and not pc_eBand4_Supp  and not pc_eBand5_Supp and
      not pc_eBand6_Supp  and not pc_eBand7_Supp  and not pc_eBand8_Supp  and not pc_eBand9_Supp  and not pc_eBand10_Supp and
      not pc_eBand11_Supp and not pc_eBand12_Supp and not pc_eBand14_Supp and not pc_eBand17_Supp and not pc_eBand18_Supp and not pc_eBand19_Supp and
      not pc_eBand20_Supp and not pc_eBand21_Supp and not pc_eBand22_Supp and not pc_eBand23_Supp and not pc_eBand24_Supp and not pc_eBand25_Supp and
      not pc_eBand33_Supp and not pc_eBand34_Supp and not pc_eBand35_Supp and not pc_eBand36_Supp and not pc_eBand37_Supp and not pc_eBand38_Supp and not pc_eBand39_Supp and
      not pc_eBand40_Supp and not pc_eBand41_Supp and not pc_eBand42_Supp and not pc_eBand43_Supp;
    // 1. mandatory PICS
    // 1.1 FGIs
    if (not pc_FeatrGrp_5)  {f_EUTRA_SetVerdictInconc(__FILE__, __LINE__, "FGI 5 set incorrectly(R9)");}
    if (not pc_FeatrGrp_6)  {f_EUTRA_SetVerdictInconc(__FILE__, __LINE__, "FGI 6 set incorrectly(R9)");}
    if (not pc_FeatrGrp_16) {f_EUTRA_SetVerdictInconc(__FILE__, __LINE__, "FGI 16 set incorrectly(R9)");}
    if (not pc_FeatrGrp_17) {f_EUTRA_SetVerdictInconc(__FILE__, __LINE__, "FGI 17 set incorrectly(R9)");}
    if (not pc_FeatrGrp_20) {f_EUTRA_SetVerdictInconc(__FILE__, __LINE__, "FGI 20 set incorrectly(R9)");}
    // 1.2 other PICS
    if (not pc_Allowed_CSG_list)                                {f_EUTRA_SetVerdictInconc(__FILE__, __LINE__, "pc_Allowed_CSG_list set incorrectly(R9)");}
    if (not pc_Manual_CSG_Selection)                            {f_EUTRA_SetVerdictInconc(__FILE__, __LINE__, "pc_Manual_CSG_Selection set incorrectly(R9)");}
    // 2. relationship between PICS
    // 2.1 relationships between FGIs - none
    // 2.2 other relationships
    if  (not pc_FeatrGrp_3 and pc_VoLTE)                        {f_EUTRA_SetVerdictInconc(__FILE__, __LINE__, "FGI 3 set incorrectly(R9)");}
    if  (not pc_FeatrGrp_7 and pc_VoLTE)                        {f_EUTRA_SetVerdictInconc(__FILE__, __LINE__, "FGI 7 set incorrectly(R9)");}
    if  (not pc_FeatrGrp_8 and pc_VoLTE)                        {f_EUTRA_SetVerdictInconc(__FILE__, __LINE__, "FGI 8 set incorrectly(R9)");}
    if  (pc_FeatrGrp_13 and v_Band13OnlySupported)              {f_EUTRA_SetVerdictInconc(__FILE__, __LINE__, "FGI 13 set incorrectly(R9)");}
    if  (pc_FeatrGrp_18 and v_Band13OnlySupported)              {f_EUTRA_SetVerdictInconc(__FILE__, __LINE__, "FGI 18 set incorrectly(R9)");}
    if  (not pc_FeatrGrp_22 and pc_UTRA)                        {f_EUTRA_SetVerdictInconc(__FILE__, __LINE__, "FGI 22 set incorrectly(R9)");}
    if  (not pc_FeatrGrp_23 and pc_Enhanced_1xCSfallback)       {f_EUTRA_SetVerdictInconc(__FILE__, __LINE__, "FGI 23 set incorrectly(R9)");}
    if  (pc_FeatrGrp_25 and v_Band13OnlySupported)              {f_EUTRA_SetVerdictInconc(__FILE__, __LINE__, "FGI 25 set incorrectly(R9)");}
    if  (not pc_FeatrGrp_26 and pc_HRPD)                        {f_EUTRA_SetVerdictInconc(__FILE__, __LINE__, "FGI 26 set incorrectly(R9)");}
    if  (pc_FeatrGrp_28 and not pc_TTI_Bundling)                {f_EUTRA_SetVerdictInconc(__FILE__, __LINE__, "pc_TTI_Bundling set incorrectly(R9)");}
    if  (pc_FeatrGrp_29 and not pc_Semi_Persistence_Scheduling) {f_EUTRA_SetVerdictInconc(__FILE__, __LINE__, "pc_Semi_Persistence_Scheduling set incorrectly(R9)");}
    if  (pc_PS_voice_centric and not pc_ROHC_profile0x0001)     {f_EUTRA_SetVerdictInconc(__FILE__, __LINE__, "pc_ROHC_profile0x0001 set incorrectly(R9)");}
    if  (pc_PS_voice_centric and not pc_ROHC_profile0x0002)     {f_EUTRA_SetVerdictInconc(__FILE__, __LINE__, "pc_ROHC_profile0x0002 set incorrectly(R9)");}
  } // fl_CheckFeatrGrpCapFromPics_r9
  /*
   * @desc      check the restrictions on FGI and other PICS as per 36.523-2 related to Release 10
   *            inconsistencies lead to verdict INCONCLUSIVE
   * @status    APPROVED
   */
  function fl_CheckFeatrGrpCapFromPics_r10()
  {
    // 1. mandatory PICS
    // 1.1 FGIs
    // 1.2 other PICS
    // 2. relationship between PICS
    // 2.1 relationships between FGIs - none
    // 2.2 other relationships
    // Empty for the time being. PICS needs to be defined first, eg UE Category 8
  } // fl_CheckFeatrGrpCapFromPics_r10
  /*
   * @desc      check the restrictions on RAT specific FGIs as prescribed in 36.331 by column FDD/ TDD diff
   *            The column FDD/ TDD diff indicates if the UE is allowed to signal different values for FDD and TDD
   *            FGIs 3,5,6,7,13,14,16,17,18,20,21,25,30 are restrained.
   *            For the time being no additional FGIs (>= 33) have restrictions.
   * @param     p_FGI_Set1
   * @param     p_FGI_Set2
   * @status    APPROVED
   */
  function fl_CheckFeatrGrpCapSplit (B32_Type p_FGI_Set1,
                                     B32_Type p_FGI_Set2)
  { // @sic R5s120250 sic@
    var IntegerList_Type v_FGI_List := {3, 5, 6, 7, 13, 14, 16, 17, 18, 20, 21, 25, 30};
    var integer v_FGI;
    var integer v_Index;
    var integer i;
    for (i:=0; i < lengthof(v_FGI_List); i:=i+1) {
      v_FGI := v_FGI_List[i];  // FGI starting with 1
      v_Index := v_FGI - 1;    // bit position, starting with 0
      if (f_Bitstring_BitIsSet(p_FGI_Set1, v_Index) != f_Bitstring_BitIsSet(p_FGI_Set2, v_Index)) {
        f_EUTRA_SetVerdictInconc(__FILE__, __LINE__, "FGI " & int2str(v_FGI) & " split erroneously");
      }
    }
  }
…
  /*
   * @desc      Fill the ue-CapabilityRAT-ContainerList in the expected template based on set PICS parameters
   *            Filled template will be used in the receiving statement in function fl_AllRatCapabilityCheck.
   * @return    template UECapabilityInformation_r8_IEs
   * @status    APPROVED
   */
  function fl_SetUeOtherSupRatCap() return template UECapabilityInformation_r8_IEs
  {
    var template (present) UE_CapabilityRAT_ContainerList v_UE_CapabilityRAT_ContainerList;
    var integer i:=0;
    if (pc_GERAN and pc_CS) {  // @sic R5s120250 sic@
      v_UE_CapabilityRAT_ContainerList[i] := cr_UE_CapabilityRAT_Container(geran_cs);
      i := i + 1;
    }
    if (pc_GERAN and pc_PS) {  // @sic R5s120250 sic@
      v_UE_CapabilityRAT_ContainerList[i] := cr_UE_CapabilityRAT_Container(geran_ps);
      i := i + 1;
    }
    if (pc_FDD or pc_TDD_LCR or pc_TDD_HCR or pc_TDD_VHCR) {
      v_UE_CapabilityRAT_ContainerList[i] := cr_UE_CapabilityRAT_Container(utra);
      i := i + 1;
    }
    if (pc_HRPD or pc_1xRTT) {
      v_UE_CapabilityRAT_ContainerList[i] := cr_UE_CapabilityRAT_Container(cdma2000_1XRTT);
      i := i + 1;
    }
    select (i) {
      case (1) {
        // do nothing
      }
      case (2) {
        v_UE_CapabilityRAT_ContainerList := {
          permutation(v_UE_CapabilityRAT_ContainerList[0], v_UE_CapabilityRAT_ContainerList[1])
        };
      }
      case (3) {
        v_UE_CapabilityRAT_ContainerList := {
          permutation(v_UE_CapabilityRAT_ContainerList[0], v_UE_CapabilityRAT_ContainerList[1], v_UE_CapabilityRAT_ContainerList[2])
        };
      }
      case (4) {
        v_UE_CapabilityRAT_ContainerList := {
          permutation(v_UE_CapabilityRAT_ContainerList[0], v_UE_CapabilityRAT_ContainerList[1], v_UE_CapabilityRAT_ContainerList[2], v_UE_CapabilityRAT_ContainerList[3])
        };
      }
      case else {
        FatalError(__FILE__, __LINE__, "invalid PICS configuration");
      }
    }
    return cr_UECapabilityInformation(v_UE_CapabilityRAT_ContainerList);
  }
…

  /*
   * @desc      Function to check some Rel 8 UE capabilities
   *            can be used for Rel 9 and Rel 10 UE capability checking as it is not changing with the release version
   *            Reference 36.523-1 Table 8.5.4.1.3.3-2
   * @param     p_ReceivedCapabilityMsg
   * @status    APPROVED
   */
  function fl_EUTRA_Check_r8(UE_EUTRA_Capability p_ReceivedCapabilityMsg) runs on EUTRA_PTC
  { // @sic R5s120250 sic@
    var UE_EUTRA_Capability v_ReceivedCapabilityMsg;
    var integer i;
    var integer v_BandEUTRA;
    var integer v_BitStringIndex;
    var bitstring v_SuppEutraBandsAccToPICS := fl_DeriveSuppEutraBandsFromPics();
    var bitstring v_SuppEutraBandsReported := int2bit(0, lengthof(v_SuppEutraBandsAccToPICS));
    v_ReceivedCapabilityMsg := p_ReceivedCapabilityMsg;
    // Check that number of supported EUTRA bands match the set PICS parameters (pc_eBanda_Supp)
    if (lengthof(v_ReceivedCapabilityMsg.rf_Parameters.supportedBandListEUTRA) != fl_GetNrOfSuppEutraBands(v_SuppEutraBandsAccToPICS)) {
      f_EUTRA_SetVerdictFailOrInconc (__FILE__, __LINE__, "Mismatch between the number of bands acc. to PICS setting and the number of bands reported by the UE");
    }
    // Check that any value B such that pc_eBandB_Supp is TRUE and different from all eutra-Band[k] where k = 1 to a - 1
    // Go through the received supportedBandListEUTRA and match against the supported bands according to the PICS
    for (i := 0; i < lengthof(v_ReceivedCapabilityMsg.rf_Parameters.supportedBandListEUTRA); i := i + 1) {
      v_BandEUTRA := v_ReceivedCapabilityMsg.rf_Parameters.supportedBandListEUTRA[i].bandEUTRA;
      v_BitStringIndex := v_BandEUTRA - 1;
      if (not f_Bitstring_BitIsSet(v_SuppEutraBandsAccToPICS, v_BitStringIndex)) {
        f_EUTRA_SetVerdictFailOrInconc (__FILE__, __LINE__, "PICS does not match reported capability for band nr. " & int2str(v_BandEUTRA) &  " or band-support has been reported more than once"); 
      } else {
        if (f_Bitstring_BitIsSet(v_SuppEutraBandsReported, v_BitStringIndex)) {
          f_EUTRA_SetVerdictFailOrInconc (__FILE__, __LINE__, "support of band nr. " & int2str(v_BandEUTRA) &  " is reported more than once"); 
        }
        v_SuppEutraBandsReported[v_BitStringIndex] := '1'B;
      }
      // NOTE:  number of bands reported by the UE and number of bands acc. to PICS setting are equal AND none of the bands is reported twice by the UE 
      //        => bands reported by the UE are the same as acc. to PICS setting (i.e. we don't need to check explicitly the PICS against the bands being reported)
    }
    // interFreqEUTRA-BandList SEQUENCE (SIZE (1..maxBands)) OF SEQUENCE {
    //* same number of entries like in supportedEUTRA-BandList
    //* Match number of entries in interFreqEUTRA-BandList with number of entries in supportedEUTRA-BandList
    for (i := 0; i < lengthof(v_ReceivedCapabilityMsg.measParameters.bandListEUTRA); i := i + 1) {
      if (lengthof(v_ReceivedCapabilityMsg.rf_Parameters.supportedBandListEUTRA) != lengthof(v_ReceivedCapabilityMsg.measParameters.bandListEUTRA[i].interFreqBandList)) {
        f_EUTRA_SetVerdictFailOrInconc (__FILE__, __LINE__, "Test Case 8.5.4.1 Step 10/18 - Number of interFreqEUTRA-BandList does not match nr. of supportedEUTRA-BandList");
      }
    }
    //@sic R5s100302 Change 6 : Correction of check of supported interRat bands sic@
    // Check that number of bands in interRAT-Parameters matches number of entries in interRAT-BandList
    // According to 36.331:
    // interRAT-BandList - One entry corresponding to each supported band of another RAT
    //                     listed in the same order as in the interRAT-Parameters.
    for (i := 0; i < lengthof(v_ReceivedCapabilityMsg.measParameters.bandListEUTRA); i := i + 1) {
      if (ispresent(v_ReceivedCapabilityMsg.measParameters.bandListEUTRA[i].interRAT_BandList)) {
        if (lengthof(v_ReceivedCapabilityMsg.measParameters.bandListEUTRA[i].interRAT_BandList) != fl_Count_Supported_interRAT_Bands(v_ReceivedCapabilityMsg)) {
          f_EUTRA_SetVerdictFailOrInconc (__FILE__, __LINE__, "Test Case 8.5.4.1 Step 10/18 - Number of listed bands in interRAT-Parameters does not match number of entires in interRAT-BandList");
        }
      }
    }
  } // fl_EUTRA_Check_r8
  /*
   * @desc      Function to match the received UECapabilityInformation with the expected UECapabilityInformation.
   *            Reference 36.523-1 Table 8.5.4.1.3.3-2
   * @param     p_ReceivedCapabilityMsg          ..  received ueCapabilitiesRAT-Container
   * @param     p_Expected_Ue_Eutra_Capability   ..  expected ueCapabilitiesRAT-Container (set by function fl_EUTRA_SetUeEutraCapability)
   * @status    APPROVED
   */
  function fl_EUTRA_MatchEutraCapMsgs_r8(UE_EUTRA_Capability p_ReceivedCapabilityMsg,
                                         template UE_EUTRA_Capability p_Expected_Ue_Eutra_Capability) runs on EUTRA_PTC
  { // @sic R5s120250 sic@
    fl_EUTRA_Check_r8(p_ReceivedCapabilityMsg);
    // Match the received decoded template with the expected template
    if (not match(p_ReceivedCapabilityMsg, p_Expected_Ue_Eutra_Capability)) {
      f_EUTRA_SetVerdictFailOrInconc (__FILE__, __LINE__, "Test Case 8.5.4.1 Step 10/18 - Received decoded template does not match with the expected template");
    }
  }
  /*
   * @desc      Function to match the received UECapabilityInformation with the expected UECapabilityInformation.
   *            Reference 36.523-1 Table 8.5.4.1.3.3-2
   * @param     p_ReceivedCapabilityMsg          ..  received ueCapabilitiesRAT-Container
   * @param     p_Expected_Ue_Eutra_Capability   ..  expected ueCapabilitiesRAT-Container (set by function fl_EUTRA_SetUeEutraCapability)
   * @status    APPROVED
   */
  function fl_EUTRA_MatchEutraCapMsgs_r9(UE_EUTRA_Capability p_ReceivedCapabilityMsg,
                                         template UE_EUTRA_Capability p_Expected_Ue_Eutra_Capability) runs on EUTRA_PTC
  { // @sic R5s120250 sic@
    var UE_EUTRA_Capability v_ReceivedCapabilityMsg;
    var UE_EUTRA_Capability_v9a0_IEs v_Ue_EUTRA_Capability_v9a0_IEs;
    var bitstring v_Ue_EUTRA_Capability_v9a0_IEsContainer;
    var integer v_Result;
    var B32_Type v_FGIsFromPICS, v_FGIsRcvd, v_CommonFGIs, v_SplitFGIsFDD, v_SplitFGIsTDD;
    var template UE_EUTRA_Capability v_Expected_Ue_Eutra_Capability;
    v_ReceivedCapabilityMsg := p_ReceivedCapabilityMsg;
    v_Expected_Ue_Eutra_Capability := p_Expected_Ue_Eutra_Capability;
    // to check some Rel 8 UE capabilities
    fl_EUTRA_Check_r8(p_ReceivedCapabilityMsg);
    // ROHC profiles not checked for Rel 8, but need to be checked here
    if (pc_PS_voice_centric and not v_ReceivedCapabilityMsg.pdcp_Parameters.supportedROHC_Profiles.profile0x0001) {
      f_EUTRA_SetVerdictInconc(__FILE__, __LINE__, "ROHC profile0x0001 not configured (R9)");
    }
    if (pc_PS_voice_centric and not v_ReceivedCapabilityMsg.pdcp_Parameters.supportedROHC_Profiles.profile0x0002) {
      f_EUTRA_SetVerdictInconc(__FILE__, __LINE__, "ROHC profile0x0002 not configured (R9)");
    }
    // the value - presence checked already - is checked as per 36.331 cl. 5.6.3.3 !!!
    // decoding of the UE_EUTRA_Capability_v9a0_IEs CONTAINER
    v_Ue_EUTRA_Capability_v9a0_IEsContainer := oct2bit(v_ReceivedCapabilityMsg.nonCriticalExtension.nonCriticalExtension.lateNonCriticalExtension);
    v_Result := decvalue (v_Ue_EUTRA_Capability_v9a0_IEsContainer, v_Ue_EUTRA_Capability_v9a0_IEs);
    if (v_Result != 0) {
      FatalError(__FILE__, __LINE__, "UE_EUTRA_Capability_v9a0_IEs cannot be decoded");
    }
    // check FGIs 33-36 featureGroupIndRel9Add-r9 if present
    if (isvalue (v_Ue_EUTRA_Capability_v9a0_IEs.featureGroupIndRel9Add_r9)) {
      // check match between PICS settings and received value
      v_FGIsFromPICS := fl_DeriveFeatrGrpCapFromPics_r9Add();
      v_FGIsRcvd := v_Ue_EUTRA_Capability_v9a0_IEs.featureGroupIndRel9Add_r9;
      if (v_FGIsFromPICS != v_FGIsRcvd) {
        f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "featureGroupIndRel9Add_r9 not consistent with PICS");
      }
    }
    else {
      // FGIs not reported: error in Rel9, all 1s in future releases assumed  (36.331 Annex B1)
      f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "featureGroupIndRel9Add_r9 must be provided in Rel 9");
    }
    // check RAT specific FGIs, if present or not
    // FGIs and Add FGIs contain common values,
    // RAT specific FGIs and Add FGIs contain potentially different RAT specific values
    // RAT specific FGIs may be omitted if they contain only values equal to the common ones
    // - FGIs 1-30 will be checked when matching
    // - FGIs 33-36 checked just before - mandatory in Rel 9
    // - if add capabilities are absent => FGIs same as the common ones
    // - if add capabilities are present => FGIs as indicated in the PICS
    // - if some add capabilities are present => mix of both
    if (pc_FDD and (pc_TDD_LCR or pc_TDD_HCR or pc_TDD_VHCR)) {
      // 1. fdd_Add_UE_EUTRA_Capabilities_r9
      if (isvalue (v_Ue_EUTRA_Capability_v9a0_IEs.fdd_Add_UE_EUTRA_Capabilities_r9)) {
        // 1.1 check consistency between Received Add FGIs and the Add PICS settings
        // 1.1.1 FGIs 1-30
        v_FGIsFromPICS := fl_DeriveFeatrGrpCapFromPics_r9_FDD();
        v_FGIsRcvd := v_Ue_EUTRA_Capability_v9a0_IEs.fdd_Add_UE_EUTRA_Capabilities_r9.featureGroupIndicators_r9;
        if (v_FGIsFromPICS != v_FGIsRcvd) {
          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "fdd_Add_UE_EUTRA_Capabilities_r9.featureGroupIndicators_r9 not consistent with PICS");
        }
        v_SplitFGIsFDD := v_FGIsRcvd; // save to later check FDD/TDD splitting
        // 1.1.2 FGIs 33-36
        v_FGIsFromPICS := fl_DeriveFeatrGrpCapFromPics_r9Add_FDD();
        v_FGIsRcvd := v_Ue_EUTRA_Capability_v9a0_IEs.fdd_Add_UE_EUTRA_Capabilities_r9.featureGroupIndRel9Add_r9;
        if (v_FGIsFromPICS != v_FGIsRcvd) {
          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "fdd_Add_UE_EUTRA_Capabilities_r9.featureGroupIndRel9Add_r9 not consistent with PICS");
        }
      }
      else {
        // 1.2 check consistency of RAT specific FGIs with the Common PICS settings
        // 1.2.1 FGIs 1-30
        v_FGIsFromPICS := fl_DeriveFeatrGrpCapFromPics_r9_FDD();
        v_CommonFGIs := fl_DeriveFeatrGrpCapFromPics();
        if (v_FGIsFromPICS != v_CommonFGIs) {
          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Common FGIs and FDD FGIs inconsistent");
        }
        v_SplitFGIsFDD := v_FGIsFromPICS; // save to later check FDD/TDD splitting
        // 1.2.2 FGIs 33-36
        v_FGIsFromPICS := fl_DeriveFeatrGrpCapFromPics_r9Add_FDD();
        v_CommonFGIs := fl_DeriveFeatrGrpCapFromPics_r9Add ();
        if (v_FGIsFromPICS != v_CommonFGIs) {
          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Common Add FGIs and FDD Add FGIs inconsistent");
        }
      }
      // 2. tdd_Add_UE_EUTRA_Capabilities_r9
      if (isvalue (v_Ue_EUTRA_Capability_v9a0_IEs.tdd_Add_UE_EUTRA_Capabilities_r9)) {
        // 2.1 check consistency between Received Add FGIs and the Add PICS settings
        // 2.1.1 FGIs 1-30
        // check match between PICS settings and received value
        v_FGIsFromPICS := fl_DeriveFeatrGrpCapFromPics_r9_TDD();
        v_FGIsRcvd := v_Ue_EUTRA_Capability_v9a0_IEs.tdd_Add_UE_EUTRA_Capabilities_r9.featureGroupIndicators_r9;
        if (v_FGIsFromPICS != v_FGIsRcvd) {
          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "tdd_Add_UE_EUTRA_Capabilities_r9.featureGroupIndicators_r9 not consistent with PICS");
        }
        v_SplitFGIsTDD := v_FGIsRcvd; // save to later check FDD/TDD splitting
        // 2.1.2 FGIs 33-36
        v_FGIsFromPICS := fl_DeriveFeatrGrpCapFromPics_r9Add_TDD();
        v_FGIsRcvd := v_Ue_EUTRA_Capability_v9a0_IEs.tdd_Add_UE_EUTRA_Capabilities_r9.featureGroupIndRel9Add_r9;
        if (v_FGIsFromPICS != v_FGIsRcvd) {
          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "tdd_Add_UE_EUTRA_Capabilities_r9.featureGroupIndRel9Add_r9 not consistent with PICS");
        }
      }
      else {
        // 2.2 check consistency of RAT specific FGIs with the Common PICS settings
        // 2.2.1. FGIs 1-30
        v_FGIsFromPICS := fl_DeriveFeatrGrpCapFromPics_r9_TDD();
        v_CommonFGIs := fl_DeriveFeatrGrpCapFromPics();
        if (v_FGIsFromPICS != v_CommonFGIs) {
          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Common FGIs and TDD FGIs inconsistent");
        }
        v_SplitFGIsTDD := v_FGIsFromPICS; // save to later check FDD/TDD splitting
        // 2.2.2 FGIs 33-36
        v_FGIsFromPICS := fl_DeriveFeatrGrpCapFromPics_r9Add_TDD();
        v_CommonFGIs := fl_DeriveFeatrGrpCapFromPics_r9Add();
        if (v_FGIsFromPICS != v_CommonFGIs) {
          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Common Add FGIs and TDD Add FGIs inconsistent");
        }
      }
      // 3. Check FGI splitting: the UE does not split where it is not allowed to signal different values for FDD and TDD
      fl_CheckFeatrGrpCapSplit (v_SplitFGIsFDD, v_SplitFGIsTDD);
    }
    // not both RATs active !!
    else { // featureGroupIndRel9Add-r9 must be provided for Rel9, RAT specific FGI information must not be provided
      // check fdd_Add_UE_EUTRA_Capabilities_r9 are absent
      if (isvalue (v_Ue_EUTRA_Capability_v9a0_IEs.fdd_Add_UE_EUTRA_Capabilities_r9)) {
        f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "fdd_Add_UE_EUTRA_Capabilities_r9 must not be provided");
      }
      // check tdd_Add_UE_EUTRA_Capabilities_r9 are absent
      if (isvalue (v_Ue_EUTRA_Capability_v9a0_IEs.tdd_Add_UE_EUTRA_Capabilities_r9)) {
        f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "tdd_Add_UE_EUTRA_Capabilities_r9 must not be provided");
      }
    }
    // Match the received decoded template with the expected template
    if ( not match ( v_ReceivedCapabilityMsg, v_Expected_Ue_Eutra_Capability )) {
      f_EUTRA_SetVerdictFailOrInconc (__FILE__, __LINE__, "Test Case 8.5.4.1 Step 10/18 - Received decoded template does not match with the expected template");
    }
  } // fl_EUTRA_MatchEutraCapMsgs_r9
  /*
   * @desc      Function to match the received UECapabilityInformation with the expected UECapabilityInformation.
   *            Reference 36.523-1 Table 8.5.4.1.3.3-2
   * @param     p_ReceivedCapabilityMsg         ..  received ueCapabilitiesRAT-Container
   * @param     p_Expected_Ue_Eutra_Capability  ..  expected ueCapabilitiesRAT-Container (set by function fl_EUTRA_SetUeEutraCapability)
   * @status    APPROVED
   */
  function fl_EUTRA_MatchEutraCapMsgs_r10(UE_EUTRA_Capability p_ReceivedCapabilityMsg,
                                          template UE_EUTRA_Capability p_Expected_Ue_Eutra_Capability) runs on EUTRA_PTC
  { // @sic R5s120250 sic@
    var UE_EUTRA_Capability v_ReceivedCapabilityMsg;
    var template UE_EUTRA_Capability v_Expected_Ue_Eutra_Capability;
    v_ReceivedCapabilityMsg := p_ReceivedCapabilityMsg;
    v_Expected_Ue_Eutra_Capability := p_Expected_Ue_Eutra_Capability;
    // to check some Rel 8 UE capabilities
    fl_EUTRA_Check_r8 (p_ReceivedCapabilityMsg);
    // ROHC profiles not checked for Rel 8, but need to be checked here
    if ((pc_PS_voice_centric) and (v_ReceivedCapabilityMsg.pdcp_Parameters.supportedROHC_Profiles.profile0x0001 == false)) {
      f_EUTRA_SetVerdictInconc(__FILE__, __LINE__, "ROHC profile0x0001 not configured (R9)");
    }
    if ((pc_PS_voice_centric) and (v_ReceivedCapabilityMsg.pdcp_Parameters.supportedROHC_Profiles.profile0x0002 == false)) {
      f_EUTRA_SetVerdictInconc(__FILE__, __LINE__, "ROHC profile0x0002 not configured (R9)");
    }
    // Rel9 to be done FFS differences to Rel9 only checking !!
    f_EUTRA_SetVerdictInconc(__FILE__, __LINE__, "Rel9 check for Rel10 UEs not yet implemented");
    // Rel10 to be done FFS
    if (v_ReceivedCapabilityMsg.accessStratumRelease == rel10) {
      if (isvalue (v_ReceivedCapabilityMsg.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension)) {
        v_Expected_Ue_Eutra_Capability.nonCriticalExtension := omit;
      }
      else {
        v_Expected_Ue_Eutra_Capability.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension := ?;
      }
    }
    // Match the received decoded template with the expected template
    if ( not match ( v_ReceivedCapabilityMsg, v_Expected_Ue_Eutra_Capability )){
      f_EUTRA_SetVerdictFailOrInconc (__FILE__, __LINE__, "Test Case 8.5.4.1 Step 10/18 - Received decoded template does not match with the expected template");
    }
  } // fl_EUTRA_MatchEutraCapMsgs_r10
…
  /*
   * @desc      Send UECapabilityEnquiry to request UE radio access capability information for EUTRA
   *            and check the received UECapabilityInformation.
   * @param     p_CellId
   * @param     p_RRC_TI
   * @status    APPROVED
   */
  function fl_EUTRA_FullCapabilityCheck(CellId_Type p_CellId,
                                        RRC_TransactionIdentifier p_RRC_TI) runs on EUTRA_PTC
  {
    var SRB_COMMON_IND v_ReceivedAspForUeCapabilityInfo;
    var UE_EUTRA_Capability v_ReceivedCapabilityMsg;
    // Fill the IE interRAT-Parameters with expected contents based on set PICS parameters
    var template UE_EUTRA_Capability v_EutraCapability := fl_EUTRA_SetExpectedUeEutraCap(cr_UeCapabilitiesRAT_Container);
    // Transmit UE Capability Enquiry message to ask for EUTRA Capability information
    SRB.send(cas_SRB1_RrcPdu_REQ(p_CellId, cs_TimingInfo_Now, cs_508_UeCapabilityEnquiry(p_RRC_TI)));
    // Receive UE Capability Information message for EUTRA capabilities and decode
    SRB.receive(car_SRB1_RrcPdu_IND(p_CellId, cr_508_UeCapabilityInformation(p_RRC_TI, cr_EutraCapInfo))) -> value v_ReceivedAspForUeCapabilityInfo;
    v_ReceivedCapabilityMsg := f_EUTRA_DecodeEutraCapMsg(v_ReceivedAspForUeCapabilityInfo.Signalling.Rrc.Dcch.message_.c1.ueCapabilityInformation.criticalExtensions.c1.ueCapabilityInformation_r8.ue_CapabilityRAT_ContainerList[0].ueCapabilityRAT_Container);
    select (v_ReceivedCapabilityMsg.accessStratumRelease) { // @sic R5s120250 sic@
      case (rel8) {
        // Check Rel 8 vs. IEs provided
        if (isvalue(v_ReceivedCapabilityMsg.nonCriticalExtension)) {
          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "UE-EUTRA-Capability-v920-IEs must not be provided by Rel 8 UE");
        }
        else {
          v_EutraCapability.nonCriticalExtension := omit;
        }
        // Match the ueCapabilitiesRAT-Container
        fl_EUTRA_MatchEutraCapMsgs_r8(v_ReceivedCapabilityMsg, v_EutraCapability);
        // Check consistency of FGI usage vs. release version as per 36.523-2 // @sic R5-120748 sic@
        // Rel 8 consistency checked unconditionally.
        fl_CheckFeatrGrpCapFromPics_r8();
      }
      case (rel9) {
        if (isvalue(v_ReceivedCapabilityMsg.nonCriticalExtension.nonCriticalExtension)) {
          // no Rel 9 (UE-EUTRA-Capability-v940-IEs) IEs provided
          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "UE_EUTRA_Capability_v940_IEs cannot omitted as Add FGIs are mandatory for Rel 9");
        }
        else {
          // Rel 9 (UE-EUTRA-Capability-v940-IEs) IEs provided !!
          if (isvalue (v_ReceivedCapabilityMsg.nonCriticalExtension.nonCriticalExtension.lateNonCriticalExtension)) {
          // Rel 9 (UE-EUTRA-Capability-v9a0-IEs) IEs provided !!
            v_EutraCapability.nonCriticalExtension.nonCriticalExtension.lateNonCriticalExtension := ?;
          }
          else {
            v_EutraCapability.nonCriticalExtension.nonCriticalExtension.lateNonCriticalExtension := omit;
          }
          if (isvalue (v_ReceivedCapabilityMsg.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension)) {
            // Rel 10 (UE-EUTRA-Capability-v1020-IEs) IEs provided !!
            f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "UE-EUTRA-Capability-v1020-IEs must not be provided by Rel 9 UE");
          }
          else {
            v_EutraCapability.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension := omit;
          }
        }
        // Match the ueCapabilitiesRAT-Container
        fl_EUTRA_MatchEutraCapMsgs_r9 ( v_ReceivedCapabilityMsg, v_EutraCapability );
        // Check consistency of FGI usage vs. release version as per 36.523-2 // @sic R5-120748 sic@
        // Rel 9 consistency checked unconditionally. Result not exploited further yet.
        fl_CheckFeatrGrpCapFromPics_r9 ();
      }
      case (rel10) {
        // Rel 9 IEs may not be provided ... FFS
        if (isvalue (v_ReceivedCapabilityMsg.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension)) {
          // no Rel 10 IEs provided
          v_EutraCapability.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension := omit;
        }
        else {
          v_EutraCapability.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension := ?;
        }
        // Match the ueCapabilitiesRAT-Container
        fl_EUTRA_MatchEutraCapMsgs_r10 ( v_ReceivedCapabilityMsg, v_EutraCapability );
        // Check consistency of FGI usage vs. release version as per 36.523-2
        // Rel 10 consistency checked unconditionally. Result not exploited further yet.
        fl_CheckFeatrGrpCapFromPics_r10 ();
      }
      case else  {
        FatalError (__FILE__, __LINE__, "Invalid AccessStratumRelease");
      }
    }
  } // fl_EUTRA_FullCapabilityCheck
  /*
   * @desc      Send UECapabilityEnquiry to request UE radio access capability information for other supported RATs and check the received UECapabilityInformation
   * @param     p_CellId
   * @param     p_RRC_TI
   * @status    APPROVED
   */
  function fl_AllRatCapabilityCheck(CellId_Type p_CellId,
                                    RRC_TransactionIdentifier p_RRC_TI) runs on EUTRA_PTC
  {
    var SRB_COMMON_IND v_ReceivedAspForUeCapabilityInfo;
    var UE_CapabilityRequest v_SuppOtherRatCapReq;
    var template UECapabilityInformation_r8_IEs v_OtherSupRatCapability := fl_SetUeOtherSupRatCap(); // Fill the UECapabilityInformation_r8_IEs with expected RAT-containers based on set PICS parameters
    var integer i:=0;
    // Fill the UECapabilityRequest IEs with requested RAT-containers based on set PICS parameters
    if (pc_GERAN and pc_CS) {
      v_SuppOtherRatCapReq[i] := geran_cs;
      i:=i+1;
    }
    if (pc_GERAN and pc_PS) {
      v_SuppOtherRatCapReq[i] := geran_ps;
      i:=i+1;
    }
    if (pc_FDD or pc_TDD_LCR or pc_TDD_HCR or pc_TDD_VHCR) {
      v_SuppOtherRatCapReq[i] := utra;
      i:=i+1;
    }
    if (pc_HRPD or pc_1xRTT) {
      v_SuppOtherRatCapReq[i] := cdma2000_1XRTT;
      i:=i+1;
    }
    // Transmit UE Capability Enquiry message for every other RAT
    SRB.send(cas_SRB1_RrcPdu_REQ(p_CellId,
                                 cs_TimingInfo_Now,
                                 cs_508_UeCapabilityEnquiry(p_RRC_TI, v_SuppOtherRatCapReq)));
    // Receive UE Capability Information message for every other supported RAT
    SRB.receive(car_SRB1_RrcPdu_IND(p_CellId, cr_508_UeCapabilityInformation(p_RRC_TI, v_OtherSupRatCapability))) -> value v_ReceivedAspForUeCapabilityInfo;
    // Decode and check the ueCapabilitiesRAT-Containers for every other supported RAT
    fl_EUTRA_CheckOtherRatCap(v_ReceivedAspForUeCapabilityInfo.Signalling.Rrc.Dcch.message_.c1.ueCapabilityInformation);
  }
...

